Economic equilibrium and optimization problems using GAMS
Notes 4: Imperfect competition and games

James R. Markusen
University of Colorado, Boulder

Monopoly

Cournot duopoly

Oligopoly with free entry and exit

Nash equilibrium with discrete strategies

M4-1 Simple partial-equilibrium monopoly model

Revenue: price times quantity, but now price is a function of quantity: $p = p(X)$, revenue = $p(X)*X$

$$Marginal\ \ revenue \;=\; \frac{d(p(X)X)}{dX} \;=\; p \;+\; X\frac{dp}{dX}$$

$$MR \;=\; p \;+\; p\frac{X}{p}\frac{dp}{dX} \;=\; p(1 \;-\; \frac{1}{\eta})$$

$$\eta \;=\; -\frac{p}{X}\frac{dX}{dp}$$ is the price elasticity of demand

$$Profits = Revenue - Cost = p(X)X - cX$$

First-order condition for profit max: MR = MC

$$\max \Pi \quad => \quad p(1 - \frac{1}{\eta}) = c$$

Special case: demand given by

$X = p^{-\sigma} M$ where M is income σ > 1

$$\frac{dX}{dp} = -\sigma p^{-\sigma - 1} M \qquad \frac{p}{X}\frac{dX}{dp} = -\sigma$$

$$MR = MC \quad => \quad p(1 - \frac{1}{\sigma}) = c$$

```
$TITLE: M4-1 simple partial equilibrium monopoly market
* consant price elasticity of demand function gives simple markup rule
* MK = 1/SI where SI (sigma in the notes) is the price elasticity of demand


PARAMETERS
 SI   price elasticity of demand
 M    income
 C    marginal cost (constant);

SI = 5;
M = 10;
C = 1;

VARIABLES
 PR   profit;

NONNEGATIVE VARIABLES
 X    output or demand
 P    price
 MK   markup;

EQUATIONS
 PROFIT   profit
 DEMAND   supply = demand
 FOC1     marginal cost >= marginal revenue using SI
 FOC2     marginal cost >= marginal revenue using variable MK
 MARKUP   markup formula;
```

```
PROFIT..  PR =E= P*X - C*X;

DEMAND..  X =E= P**(-SI)*M;

FOC1..    C =G= P*(1-1/SI);

FOC2..    C =G= P*(1-MK);

MARKUP..  MK =G= 1/SI;

PR.L = 1;
P.L = 1.25;
X.L = 3;

MODEL PMAXNLP /PROFIT, DEMAND/;
SOLVE PMAXNLP MAXIMIZING PR USING NLP;

MODEL PMAXMCP1 /FOC1.X, DEMAND.P/;
SOLVE PMAXMCP1 USING MCP;

MODEL PMAXMCP2 /FOC2.X, DEMAND.P, MARKUP.MK /;
SOLVE PMAXMCP2 USING MCP;

PARAMETERS
 CSMONO   consumer surplus under monopoly
 PRMONO   profits under monopoly
 WMONO    welfare under monopoly
 CSCOMP   consumer surplus under competition
 PRCOMP   profits under competition
 WCOMP    welfare under competition;
```

```
CSMONO = 1/(SI-1)*P.L*X.L;
PRMONO = P.L*X.L - C*X.L;
WMONO  = CSMONO + PRMONO;

* compare to the competitive solution by constraining MK = 0;

MK.FX = 0;

SOLVE PMAXMCP2 USING MCP;

CSCOMP = 1/(SI-1)*P.L*X.L;
PRCOMP = P.L*X.L - C*X.L;
WCOMP= CSCOMP + PRCOMP;

DISPLAY CSMONO, PRMONO, WMONO;
DISPLAY CSCOMP, PRCOMP, WCOMP;


$ontext
example showing point from economics of regulation
Suppose that there is a fixed cost to the firm FC
Then the competitive solution means that the firm is making losses
 but the competitive solution is still socially optimal
First best policy is marginal cost pricing with a subsidy
$offtext

PARAMETERS
 FC   fixed cost /0.5/;
```

```
MK.UP = +INF;
MK.LO = 0;

SOLVE PMAXMCP2 USING MCP;

CSMONO = 1/(SI-1)*P.L*X.L;
PRMONO = P.L*X.L - C*X.L - FC;
WMONO  = CSMONO + PRMONO;

* compare to the competitive solution by constraining MK = 0;

MK.FX = 0;

SOLVE PMAXMCP2 USING MCP;

CSCOMP = 1/(SI-1)*P.L*X.L;
PRCOMP = P.L*X.L - C*X.L - FC;
WCOMP= CSCOMP + PRCOMP;

DISPLAY CSMONO, PRMONO, WMONO;
DISPLAY CSCOMP, PRCOMP, WCOMP;
```

M4-2 Partial-equilibrium oligopoly model with free entry and exit

Firms have a cost function that has a constant marginal cost c and a fixed cost f.

Marginal cost in units of labor is denoted by *mc* and total cost (*tc*) and average cost (*ac*) for an *X* firm are as follows:

$$tc = cX + f \qquad ac = \frac{tc}{X} = c + \frac{f}{X} \qquad mc = c$$

Auto industry: Minimum efficient scale, thousands of units per year

Foundry	1500
Pressing	1000
Powertrain	500
Final assembly	300


Cost
$ac = mc + \frac{fc}{X}$
mc
X

Cost penalties from sub-optimal scale

Level of production	50,000	100,000	200,000	400,000	800,000
Cost penalty (%)	20	10-15	3-5	0	-1

Size of plant (% of MES)	100	80	60	30	10
Cost penalty (%)	0	3	6.8	19.5	34.5

Cournot-Nash (or Cournot for short) competition in which firms pick a quantity as a best response to their rivals' quantities.

Revenue for a Cournot firm i and selling in country j is given by the price times quantity of the firm's sales. Price is a function of all firms' sales.

US automobile production 2005 (excludes firms exclusively producing trucks)

Number		Market share	Cummulative market share
3382315	General Motors	0.288	0.288
2965872	Ford	0.252	0.540
1652703	Chrysler	0.141	0.681
1283829	Toyota	0.109	0.790
973290	Honda	0.083	0.873
835946	Nissan	0.071	0.944
251147	Daimler Benz	0.021	0.965
125086	BMW	0.011	0.976
122328	Fuji (Subaru)	0.010	0.986
75200	Mazda	0.006	0.993
88003	Mitsubishi	0.007	1.000
11755719			

$R_i = p(X)X_i$.   *X* is total sales: $X = \sum_i X_i$

Cournot conjectures imply that $\partial X/\partial X_i = 1$; a one-unit increase in the firm's own supply is a one-unit increase in market supply.

Marginal revenue is then

$$\frac{\partial R_i}{\partial X_i} = p + X_i \frac{\partial p}{\partial X}\frac{\partial X}{\partial X_i} = p + X_i \frac{\partial p}{\partial X} \qquad \text{since } \frac{\partial X}{\partial X_i} = 1$$

Now multiple and divide the right-hand equation by total market supply and also by the price.

$$\frac{\partial R_i}{\partial X_i} = p + X_i \frac{\partial p}{\partial X} = p + p\frac{X_i}{X}\left[\frac{X}{p}\frac{\partial p}{\partial X}\right]$$

The term in square brackets in is just the inverse of the price elasticity of demand.

$$\frac{\partial R_i}{\partial X_i} = p\left[1 - \frac{X_i}{X}\frac{1}{\sigma}\right] \qquad \sigma \equiv -\left[\frac{p}{X}\frac{\partial X}{\partial p}\right] \qquad (elast\ of\ demand)$$

The term X_{ij}/X_j in (11.6) is just firm i's market share in market j, which we can denote by s_{ij} .

$$mr_i = p\left[1 - \frac{s_i}{\sigma}\right] = mc_i \qquad mr_i = p\left[1 - \frac{1}{N\sigma}\right] = mc_i$$

If all firms are identical, then each firm's market share is just 1/N where N is the number of firms in equilibrium.

```
NONNEGATIVE VARIABLES
 X    output or demand
 P    price
 MK   markup
 N    number of firms in equilibrium;

EQUATIONS
 DEMAND   supply = demand
 PRICING  marginal cost >= marginal revenue using variable MK
 MARKUP   markup formula
 ZEROPROF zero profits;

DEMAND..   N*X =E= P**(-SI)*M;

PRICING..  C =G= P*(1-MK);

MARKUP..   MK =G= 1/(N*SI);

ZEROPROF.. 0 =G= P*X - C*X - FC;

P.L = 1; X.L = 10; N.L = 2.5; MK.L = 1/(N.L*SI);


MODEL FREEENT /DEMAND.P, PRICING.X, MARKUP.MK, ZEROPROF.N /;
SOLVE FREEENT USING MCP;
```

Counterfactual: double the size of the economy: M = 50.

This creates a welfare gain (per capita) that would not be present in a competitive model or the monopoly model.

- output per firm rises, firm’s become more efficient
- thus average cost = price falls, which is a measure of efficiency or productivity
- the markup rate falls, indicating a smaller difference between price and marginal cost (p = mc is required for first best)

4.3 Cournot and Bertrand oligopoly with continuous strategies

Two firms h and f (as in countries h and f) produce imperfect substitutes for the world market:

(a) linear inverse demand curve for each good
(b) each firm has a constant marginal cost
(c) fixed costs are ignored.

$$p_h = \alpha - \beta X_h - \gamma X_f \quad p_f = \alpha - \beta X_f - \gamma X_h \qquad \beta \geq \gamma$$

$$\pi_i = p_h X_i - c_h X_i = (\alpha - \beta X_i - \gamma X_j) X_i - c_h X_i \qquad i \neq j$$

Cournot Nash competition is the behavioral assumption that each firm maximizes its profits treating their rival's output as fixed. (Best response.)

Best response Cournot-Nash equilibrium is the solution to the two first-order conditions for h and f

$$c_h \geq \alpha - 2\beta X_h - \gamma X_f \qquad\qquad c_f \geq \alpha - 2\beta X_f - \gamma X_h$$

These FOC are commonly referred to as “best response” or “reaction” functions. Here they can be rewritten as:

$$X_h = \frac{(\alpha - c_h)}{2\beta} - \frac{\gamma}{2\beta}X_f \qquad X_f = \frac{(\alpha - c_f)}{2\beta} - \frac{\gamma}{2\beta}X_h$$

They can be solved explicitly, easy in symmetric case with identical marginal costs (the two outputs are then identical):

$$X_i = \frac{\alpha - c}{2\beta + \gamma}$$

Figure 20.1



Figure 20.2



```
$TITLE: M4-3 James Markusen, University of Colorado, Boulder
* Cournot with continuous strategies

$ONTEXT
begin with Cournot doupoly
single unified market, constant marginal costs
goods XH and XF are imperfect substitutes
inverse demand functions PH = ALPHA - BETA*XH - GAMMA*XF   BETA > GAMMA

maximizing profits gives FOC (implicity reaction functions)
PROFIT = PH*XH - CH*XH = (ALPHA - BETA*XH - GAMMA*XF)*XH - CH*XH

first order condition:  ALPHA - 2*BETA*XH - GAMMA*XF - CH = 0
$OFFTEXT


PARAMETERS
 ALPHA   intercept of demand curve
 BETA    slope of inverse demand curve wrt own quantity
 GAMMA   slope of inverse demand curve wrt rival's quantity
 CH      marginal cost of home firm
 CF      marginal cost of foreign firm
 RESULTS(*,*);

ALPHA = 12;
BETA = 2;
GAMMA = 1.5;
CH = 1;
CF = 1;
```

```
VARIABLES
 PROFH       profit of firm h
 PROFF       profit of firm f;

NONNEGATIVE VARIABLES
 PH          price of XH
 PF          price of XF
 XH          quantity of XH
 XF          quantity of XF;

EQUATIONS
 PROFITH     profit of firm h
 PROFITF     profit of firm f
 PRICEH      inverse demand curve facing firm h
 PRICEF      inverse demand curve facing firm f
 HCOURNOT    cournot FOC for firm h (reaction function)
 FCOURNOT    cournot FOC for firm f (reaction function);

PROFITH..   PROFH =E= PH*XH - CH*XH;

PROFITF..   PROFF =E= PF*XF - CF*XF;

PRICEH..    PH =E= ALPHA - BETA*XH - GAMMA*XF;

PRICEF..    PF =E= ALPHA - BETA*XF - GAMMA*XH;

HCOURNOT..  CH =G= ALPHA - 2*BETA*XH - GAMMA*XF;

FCOURNOT..  CF =G= ALPHA - 2*BETA*XF - GAMMA*XH;
```

```
* actually only need the two first-order conditions to solve and then
* back out other variables, but harmless to include the other variables

MODEL COURNOT  /HCOURNOT.XH, FCOURNOT.XF,
                priceh.ph, pricef.pf, profith.profh, profitf.proff/;

SOLVE COURNOT USING MCP;

RESULTS("PROFH", "COURNOT") = PROFH.L;
RESULTS("PROFF", "COURNOT") = PROFF.L;
RESULTS("XH", "COURNOT") = XH.L;
RESULTS("XF", "COURNOT") = XF.L;
RESULTS("PH", "COURNOT") = PH.L;
RESULTS("PF", "COURNOT") = PF.L;


* solve for Cournot equilibrium using nlp via "diagonalization"
* max profits for H holding XF constant
* free up XF, hold XH at its solution value, max profits for F
* free up XH, hold XF at its solution value, max profits for H
* repeat


SETS I /I1*I10/;

MODEL COURNOTNLP /PROFITH, PROFITF, PRICEH, PRICEF/;

XH.L = 1; XF.L = 1; PH.L = 1; PF.L = 1;
```

```
LOOP(I,

XH.UP = +INF; XH.LO = 0;
XF.FX = XF.L;

SOLVE COURNOTNLP USING NLP MAXIMIZING PROFH;

XF.UP = +INF; XF.LO = 0;
XH.FX = XH.L;

SOLVE COURNOTNLP USING NLP MAXIMIZING PROFF;

);


* solve for collusive outcome
XH.UP = +INF; XH.LO = 0; XF.UP = +INF; XF.LO = 0;

VARIABLES
 JPROF    joint profits payoff;
EQUATIONS
 JPROFIT  joint profit function;

JPROFIT..  JPROF =E= PROFH + PROFF;

MODEL COLLUSION /JPROFIT, PROFITH, PROFITF, PRICEH, PRICEF/;

SOLVE COLLUSION USING NLP MAXIMIZING JPROF;

RESULTS("PROFH", "JMAX") = PROFH.L;
```

```
RESULTS("PROFF", "JMAX") = PROFF.L;
RESULTS("XH", "JMAX") = XH.L;
RESULTS("XF", "JMAX") = XF.L;
RESULTS("PH", "JMAX") = PH.L;
RESULTS("PF", "JMAX") = PF.L;


* solve for the competitive outcome
* add two equations for price equals marginal cost, drop reaction functions

EQUATIONS
 COMPH   price equals marginal cost for XH
 COMPF   price equals marginal cost for XF;

COMPH..  CH =G= PH;
COMPF..  CF =G= PF;

MODEL COMP /PROFITH.PROFH, PROFITF.PROFF, PRICEH.XH, PRICEF.XF,
            COMPH.PH, COMPF.PF/;

SOLVE COMP USING MCP;

RESULTS("PROFH", "COMP") = PROFH.L;
RESULTS("PROFF", "COMP") = PROFF.L;
RESULTS("XH", "COMP") = XH.L;
RESULTS("XF", "COMP") = XF.L;
RESULTS("PH", "COMP") = PH.L;
RESULTS("PF", "COMP") = PF.L;

DISPLAY RESULTS;
```

towards emerging economies.

ESTIMATED FULL COST OF BRINGING A NEW CHEMICAL OR BIOLOGICAL ENTITY TO MARKET ($ MILLION - YEAR 2011 $)



Source: J. Mestre-Ferrandiz, J. Sussex and A. Towse, The R&D cost of a new medicine, Office of Health Economics, December 2012 (Hansen, 1979; Wiggins, 1987; DiMasi et al, 1991; OTA, 1993; DiMasi et al, 2003; Mestre-Ferrandiz et al, 2012)

4.4 Nash equilibria with discrete strategies

Gams has some great set features that allow a modeler to capture lots of very interesting economics.

Here, I present a simple example of a two-player normal-form game in which each player has three strategies.

This particular version is motivated by a two-country trade model with multinational firms in which there is one firm in each country. Each firm may:

not enter, strategy 0
enter with a single plant at home, exporting to the other country, strategy 1
enter with plants in both countries, serving each market from a local plant, strategy 2

In an actual model, the numerical values in the payoff matrices are solved for from the underlying duopoly problem. Here I'll just make up number consistent with the underlying example.

```
SETS R  strategies for firm h /SH0, SH1, SH2/
     C  strategies for firm f /SF0, SF1, SF2/;

ALIAS(R,RR)
ALIAS(C,CC);

TABLE PAYOFFH(*,*)
      SF0   SF1   SF2
SH0   -.1   -.1   -.1
SH1    10     6     3
SH2    12     5     2 ;

TABLE PAYOFFF(*,*)
      SF0   SF1   SF2
SH0   -.1    10    12
SH1   -.1     6     5
SH2   -.1     3     2 ;
```

A best response Nash equilibrium, involves player h picking the row element that is the largest given the column selected by player f and vice versa (f picks the highest column for h’s row pick).

There is GAMS command that identifies the best response strategy. First, some GAMS notation.

X = 1$(Y EQ 1) means:

“set X equal to one if Y is equal to one, otherwise set X = 0"

X$(Y EQ 1) = 1 means something subtlely different:

“set X equal to one if Y is equal to one, otherwise leave the existing value of X unchanged”

We will need the first version here.

Let ROWMAX(R,C) be a matrix for h.

A value = 1 in cell (R,C) will denote h's best response row R when f plays column strategy C. Non-optimal responses = 0.

Let COLMAX(R,C) be a matrix for f.

A value = 1 in cell (R,C) will denote f's best response column C when h plays row strategy R. Non-optimal responses = 0.

The crucial GAMS command is SMAX (set max):

SMAX(RR, PAYOFFH(RR,C)) is the maximum value of the parameter PAYOFFH over the rows, for a given column C

The best-response matrices with zeros and ones are given by:

```
ROWMAX(R,C) = 1$(PAYOFFH(R,C) EQ SMAX(RR, PAYOFFH(RR,C)));

COLMAX(R,C) = 1$(PAYOFFF(R,C) EQ SMAX(CC, PAYOFFF(R,CC)));
```

Now multiple these two matrices together element by element, to get a new matrix NASHEQ(R,C).

A one denote a best response for both h and f, and hence that (R,C) cell is a Nash equilibrium.

```
NASHEQ(R,C) = ROWMAX(R,C)*COLMAX(R,C);
```

Finally, the profits at each Nash equilibrium are given by

```
PROFHNE(R,C) = PAYOFFH(R,C)$NASHEQ(R,C);

PROFFNE(R,C) = PAYOFFF(R,C)$NASHEQ(R,C);
```

This technique will find ALL pure-strategy Nash equilibria. The second example shows a case of multiple equilibria.

Case 1: each firm chooses one plant and exports to the other country (1,1), exporting duopoly shown above

Case 2: three equilibria. Exporting duopoly as in Case 1, or one firm chooses two plants, and the other firm does not enter: (1,1), (2,0), (0,2)

Case 3: each firm chooses two plants, a horizontal multinational duopoly (2,2)

4.5 Networks and logistics

This is a proto-typical model of an common operations research problem.

In this example, there are three production plants and three markets.

(plant locations and markets are distinct, but that is not important to the problem)

```
SETS
  I plants  /GUANGDONG, HERMOSILLO, BILOXI/
  J markets /NEW-YORK, CHICAGO, DENVER/;
```

In the first simple example, plants have fixed capacity and markets have a fixed demand (capacity must be GE to demand or GAMS returns “infeasible” as a solution.

```
PARAMETERS
 A(I)    plant capacity /GUANGDONG 4, HERMOSILLO 3, BILOXI 4/
 B(J)    market size j  /NEW-YORK 3, CHICAGO 2, DENVER 1/
 C(I)    plant marginal cost of production /GUANGDONG 1,
            HERMOSILLO 1, BILOXI 3/
 T(I,J)  transport cost rate from market i to j
 F       freight rate parameter /90/;
```

Distance between plants and markets is crucial. Here is how to declare and assign a two-dimension parameter in GAMS.

```
TABLE DIST(I,J)  distance
            NEW-YORK    CHICAGO    DENVER
GUANGDONG     9           8          7
HERMOSILLO    4           2         1.5
BILOXI        2           2          3;
```

The following allows distance to be converted to costs.

```
T(I,J) = F*DIST(I,J)/500;
```

Here are the variables and equations.

```
VARIABLES
 COST   variable cost to be minimized;

NONNEGATIVE VARIABLES
 X(I,J) shipment from i to j;

EQUATIONS
 SUPPLY(I)  supply constraint
 DEMAND(J)  demand constraint
 OBJDEF     objective function to be minimized;

SUPPLY(I)..  A(I) =G= SUM(J, X(I,J));

DEMAND(J)..  SUM(I, X(I,J)) =G= B(J);

OBJDEF..     COST =E= SUM((I,J), X(I,J)*(C(I) + T(I,J)));

MODEL MNLP /SUPPLY, DEMAND, OBJDEF/;
  X.L(I,J) = 1;
SOLVE MNLP USING NLP MINIMIZING COST;
```

```
---- VAR X  shipment from i to j

                          LOWER      LEVEL      UPPER     MARGINAL

GUANGDONG .NEW-YORK         .        3.000      +INF         .
GUANGDONG .CHICAGO          .          .        +INF       0.180
GUANGDONG .DENVER           .          .        +INF       0.090
HERMOSILLO.NEW-YORK         .          .        +INF         .
HERMOSILLO.CHICAGO          .        2.000      +INF         .
HERMOSILLO.DENVER           .        1.000      +INF         .
BILOXI    .NEW-YORK         .          .        +INF       0.740
BILOXI    .CHICAGO          .          .        +INF       1.100
BILOXI    .DENVER           .          .        +INF       1.370
```

Interpretation of the marginal 0.740 in an NLP program

Cost of sending 1 unit from Biloxi to NY: 0.360 + 3.000 = 3.360
Saving from not sending 1 unit from Guangdong to NY: 1.620 + 1 = 2.620.

3.360 - 2.620 = 0.740 increase in total cost of serving NY.

Counterfactual: raise demand in Chicago to 5.

B("CHICAGO") = 5;

```
---- VAR X  shipment from i to j

                          LOWER     LEVEL     UPPER    MARGINAL

GUANGDONG .NEW-YORK          .       1.000     +INF       .
GUANGDONG .CHICAGO           .       2.000     +INF       .
GUANGDONG .DENVER            .       1.000     +INF       .
HERMOSILLO.NEW-YORK          .         .       +INF      0.180
HERMOSILLO.CHICAGO           .       3.000     +INF       .
HERMOSILLO.DENVER            .         .       +INF      0.090
BILOXI    .NEW-YORK          .       2.000     +INF       .
BILOXI    .CHICAGO           .         .       +INF      0.180
BILOXI    .DENVER            .         .       +INF      0.540
```

M4-5b add demand functions in each market.

inverse demand functions in market j are given by

$$P(J) = 4 - D(J)/B(J)$$

where P is price, D is demand, and
B (parameter) is market size in J:

Doubling B means demand D doubles holding price constant.

Revenue and marginal revenue in market j are given by

revenue j = $P(J)D(J) = (4 - D(J)/B(J))D(J)$

*MR(J) = 4 - 2*D(J)/B(J)* $MR(J) = 4 - 2*D(J)/B(J)$

(MR not used in NLP version M4-5b)

```
$TITLE: M4-5b James Markusen, University of Colorado, Boulder
* Multi-market network, logistics, NLP version
* fixed plant capacities, adds market demand function to M4-5a
* production locations differ in marginal costs, distance to markets
$ontext
inverse demand functions in market j are given by

  P(J) = 4 - D(J)/B(J) where P is price, D is demand, and B (parameter)

  is market size in J: doubling B means D doubles holding price constant.
Revenue and marginal revenue in market j are given by
  (MR not used in MCP version)

 revenue j = P(J)*D(J) = (4 - D(J)/B(J))*D(J)   MR(J) = 4 - 2*D(J)/B(J)
$offtext

SETS
 I plants  /GUANGDONG, HERMOSILLO, BILOXI/
 J markets /NEW-YORK, CHICAGO, DENVER/;

PARAMETERS
 A(I)    plant capacity /GUANGDONG 4, HERMOSILLO 3, BILOXI 4/
 B(J)    market size j  /NEW-YORK 12, CHICAGO 8, DENVER 4/
 C(I)    plant marginal cost of production /GUANGDONG 1, HERMOSILLO 1, BILOXI 3/
 T(I,J)  transport cost rate from market i to j
 F       freight rate parameter /90/;

TABLE DIST(I,J)  distance
           NEW-YORK    CHICAGO    DENVER
GUANGDONG    9           8          7
HERMOSILLO   4           2         1.5
BILOXI       2           2          3;
```

```
DISPLAY DIST;

T(I,J) = F*DIST(I,J)/500;
DISPLAY C;

VARIABLES
 PROF   variable profit to be maximized;

NONNEGATIVE VARIABLES
 X(I,J) shipment from i to j
 D(J)   demand in market j
 LAMBDA shadow price on capacity constraint at plant I;

EQUATIONS
 SUPPLY(I)  supply constraint
 DEMAND(J)  demand constraint
 PROFIT     objective function;

SUPPLY(I)..  A(I) =G= SUM(J, X(I,J));

DEMAND(J)..  SUM(I, X(I,J)) =G= D(J);

PROFIT..     PROF =E= SUM(J, (4 - D(J)/B(J))*D(J))
                    - SUM((I,J), (C(I) + T(I,J))*X(I,J));

MODEL MNLP /PROFIT, SUPPLY, DEMAND/;

X.L(I,J) = 1;
D.L(J) = 1;

SOLVE MNLP USING NLP MAXIMIZING PROF;
```

```
B("CHICAGO") = 20;
SOLVE MNLP USING NLP MAXIMIZING PROF;
```

```
$TITLE: M4-5c James Markusen, University of Colorado, Boulder
* Multi-market network, logistics, MCP version
* fixed plant capacities, adds market demand function to M4-5a
* production locations differ in marginal costs, distance to markets
$ontext
inverse demand functions in market j are given by

  P(J) = 4 - D(J)/B(J) where P is price, D is demand, and B (parameter)

  is market size in J: doubling B means D doubles holding price constant.
Revenue and marginal revenue in market j are given by

 revenue j = P(J)*D(J) = (4 - D(J)/B(J))*D(J)   MR(J) = 4 - 2*D(J)/B(J)
$offtext

SETS
 I plants  /GUANGDONG, HERMOSILLO, BILOXI/
 J markets /NEW-YORK, CHICAGO, DENVER/;

PARAMETERS
 A(I)    plant capacity /GUANGDONG 4, HERMOSILLO 3, BILOXI 4/
 B(J)    market size j  /NEW-YORK 12, CHICAGO 8, DENVER 4/
 C(I)    plant marginal cost of production /GUANGDONG 1, HERMOSILLO 1, BILOXI 3/
 T(I,J)  transport cost rate from market i to j
 F       freight rate parameter /90/
 PROFIT  profit - extracted after solve;

TABLE DIST(I,J)  distance
           NEW-YORK    CHICAGO    DENVER
GUANGDONG    9          8          7
HERMOSILLO   4          2         1.5
BILOXI       2          2          3;
```

```
DISPLAY DIST;

T(I,J) = F*DIST(I,J)/500;
DISPLAY C;

NONNEGATIVE VARIABLES
 X(I,J)    shipment from i to j
 D(J)      demand in market j
 LAMBDA(I) shadow price on capacity constraint at plant I;

EQUATIONS
 SUPPLY(I)  supply constraint
 DEMAND(J)  demand constraint
 FOC(I,J)   first order condition for X(I J) MC GE MR;

SUPPLY(I)..  A(I) =G= SUM(J, X(I,J));

DEMAND(J)..  SUM(I, X(I,J)) =G= D(J);

FOC(I,J)..   C(I) + T(I,J) + LAMBDA(I) =G= 4 - 2*D(J)/B(J);

MODEL MMCP /SUPPLY.LAMBDA, DEMAND.D, FOC.X/;

X.L(I,J) = 1;
D.L(J) = 1;

SOLVE MMCP USING MCP;

PROFIT = SUM(J, (4 - D.L(J)/B(J))*D.L(J))
                   - SUM((I,J), (C(I) + T(I,J))*X.L(I,J));
DISPLAY PROFIT;
```

```
B("CHICAGO") = 20;
SOLVE MMCP USING MCP;

PROFIT = SUM(J, (4 - D.L(J)/B(J))*D.L(J))
                    - SUM((I,J), (C(I) + T(I,J))*X.L(I,J));
DISPLAY PROFIT;
```

```
---- VAR X  shipment from i to j   (NLP version)

                         LOWER     LEVEL     UPPER    MARGINAL

GUANGDONG .NEW-YORK        .        1.720     +INF        .
GUANGDONG .CHICAGO         .        0.280     +INF        .
GUANGDONG .DENVER          .        2.000     +INF        .
HERMOSILLO.NEW-YORK        .          .       +INF      -0.180
HERMOSILLO.CHICAGO         .        3.000     +INF        .
HERMOSILLO.DENVER          .          .       +INF      -0.090
BILOXI    .NEW-YORK        .        2.120     +INF        EPS
BILOXI    .CHICAGO         .          .       +INF      -0.180
BILOXI    .DENVER          .          .       +INF      -0.540


---- VAR D  demand in market j

               LOWER     LEVEL     UPPER    MARGINAL

NEW-YORK         .        3.840     +INF       EPS
CHICAGO          .        3.280     +INF       EPS
DENVER           .        2.000     +INF        .
```

Counterfactual: make Chicago bigger (NLP version)

---- VAR X shipment from i to j

	LOWER	LEVEL	UPPER	MARGINAL
GUANGDONG .NEW-YORK	.	.	+INF	-0.180
GUANGDONG .CHICAGO	.	2.458	+INF	EPS
GUANGDONG .DENVER	.	1.542	+INF	.
HERMOSILLO.NEW-YORK	.	.	+INF	-0.360
HERMOSILLO.CHICAGO	.	3.000	+INF	.
HERMOSILLO.DENVER	.	.	+INF	-0.090
BILOXI .NEW-YORK	.	3.547	+INF	.
BILOXI .CHICAGO	.	0.453	+INF	EPS
BILOXI .DENVER	.	.	+INF	-0.360

---- VAR D demand in market j

	LOWER	LEVEL	UPPER	MARGINAL
NEW-YORK	.	3.547	+INF	.
CHICAGO	.	5.911	+INF	.
DENVER	.	1.542	+INF	.

```
---- VAR X  shipment from i to j        (MCP version)

                          LOWER     LEVEL     UPPER    MARGINAL
GUANGDONG .NEW-YORK         .       1.720     +INF        .
GUANGDONG .CHICAGO          .       0.280     +INF        .
GUANGDONG .DENVER           .       2.000     +INF        .
HERMOSILLO.NEW-YORK         .         .       +INF      0.180
HERMOSILLO.CHICAGO          .       3.000     +INF        .
HERMOSILLO.DENVER           .         .       +INF      0.090
BILOXI    .NEW-YORK         .       2.120     +INF        .
BILOXI    .CHICAGO          .         .       +INF      0.180
BILOXI    .DENVER           .         .       +INF      0.540

---- VAR D  demand in market j

               LOWER     LEVEL     UPPER    MARGINAL
NEW-YORK         .       3.840     +INF        .
CHICAGO          .       3.280     +INF        .
DENVER           .       2.000     +INF        .

---- VAR LAMBDA  shadow price on capacity constraint at plant I

                 LOWER     LEVEL     UPPER    MARGINAL
GUANGDONG          .       0.740     +INF        .
HERMOSILLO         .       1.820     +INF        .
BILOXI             .         .       +INF      1.880
```

---- VAR X shipment from i to j   (Chcago biger, MCP version)

	LOWER	LEVEL	UPPER	MARGINAL
GUANGDONG .NEW-YORK	.	.	+INF	0.180
GUANGDONG .CHICAGO	.	2.458	+INF	.
GUANGDONG .DENVER	.	1.542	+INF	.
HERMOSILLO.NEW-YORK	.	.	+INF	0.360
HERMOSILLO.CHICAGO	.	3.000	+INF	.
HERMOSILLO.DENVER	.	.	+INF	0.090
BILOXI .NEW-YORK	.	3.547	+INF	.
BILOXI .CHICAGO	.	0.453	+INF	.
BILOXI .DENVER	.	.	+INF	0.360

---- VAR D demand in market j

	LOWER	LEVEL	UPPER	MARGINAL
NEW-YORK	.	3.547	+INF	.
CHICAGO	.	5.911	+INF	.
DENVER	.	1.542	+INF	.

---- VAR LAMBDA shadow price on capacity constraint at plant I

	LOWER	LEVEL	UPPER	MARGINAL
GUANGDONG	.	0.969	+INF	.
HERMOSILLO	.	2.049	+INF	.
BILOXI	.	0.049	+INF	.

Exercise 5 Refinery scheduling problem

A refinery has one input, crude oil (CO), and produces 3 outputs:

Gasoline
Diesel
Kerosene

The technology is call a Constant Elasticity of Transformation (CET) function, producing multiple outputs from one input.

$$\left(\sum_i \alpha_i \left(\frac{X_i}{\alpha_i} \right)^{\beta} \right)^{\frac{1}{\beta}} = CO \qquad \infty \geq \beta \geq 1, \qquad \sigma = \frac{1}{\beta - 1}$$

Note that if β = 2, for example, this is just the equation of a circle. This special case has an elasticity of transformation σ = 1.

If there are only two outputs, we would simply call this the PPF.

Let p_i denote the price of product i. The "unit revenue function" is a value function, the *maximum* revenue from one unit of input.

$$r(p) = \underset{x}{Max} \sum p_i x_i + \lambda \left[\left(\sum_i \alpha_i \left(\frac{x_i}{\alpha_i} \right)^{\beta} \right)^{\frac{1}{\beta}} - 1 \right]$$

If you do the algebra, this unit value function is given by

$$r(p) = \left(\sum \alpha_i p_i^{\sigma + 1} \right)^{\frac{1}{\sigma + 1}} \qquad R(p) = \left(\sum \alpha_i p_i^{\sigma + 1} \right)^{\frac{1}{\sigma + 1}} CO$$

where *R(p)* is the total revenue derived from *CO* units of input. Applying Shepard's lemma to *R(p)*, optimal outputs are

$$\frac{\partial R(p)}{\partial p_i} = X_i = ????$$

Exercise 5

(A) given the revenue function, apply Shepard's lemma to get the optimal supply functions for the three products

(B) solve for optimal product outputs using exercise-q5.gms

NLP formulation

MCP formulation